(última atualização em 08/11/2008)

UNIVERSIDADE FEDERAL DO RIO GRANDE DO SUL

INSTITUTO DE MATEMÁTICA
DEPARTAMENTO DE MATEMÁTICA PURA E APLICADA
Disciplina: MAT01168 -Matemática Aplicada -Semestre Letivo 2008/2
Professoras: Elisabeta Gallicchio e Irene Strauch

TERCEIRA ÁREA
SÉRIES DE FOURIER UTILIZANDO MAPLE

APLICAÇÃO: SISTEMA MASSA MOLA SUBMETIDO A UMA FORÇA EXTERNA PERIÓDICA

$$\frac{m\,d^2x}{dt^2}+\frac{c\,dx}{dt}+kx=f(t),\ \textbf{com f(t)=f(t+T)}$$

Solução do problema:

A resolução do problema será feita em cinco etapas, a saber:
Parte1: a)Cálculo da Série de Fourier da Força Externa (input) ; b) A série harmônica (forma compacta) da força; c) O espectro de freqüência da força.
Parte 2: Substituição da Força Externa na EDO.
Parte 3: Solução da EDOLHomogênea associada x_h
Parte 4: Uso do Método dos Coeficientes a Determinar para o cálculo da resposta permanente x_p.
Parte 5: Espectro de Amplitude da resposta permanente.

Parte1: a) Cálculo da Série de Fourier da Força Externa (input).

```
> restart:with(DEtools):with(plots):with(plottools):Digits:=3;
  assume(n,integer):assume(p,integer):interface(showassumed=0)
  :#Para não exibir n~ e p~
```

```
Warning, the name changecoords has been redefined
```
(1.1)

```
Warning, the names arrow and translate have been redefined
```

$$Digits := 3$$

```
> edo:=m*diff(x(t),t$2)+c*diff(x(t),t)+k*x(t)=f(t);
```

$$edo := m\left(\frac{d^2}{dt^2}x(t)\right)+c\left(\frac{d}{dt}x(t)\right)+k\,x(t)=f(t) \qquad \textbf{(1.2)}$$

Dá-se entrada ao período e calcula-se a freqüência fundamental

```
> T:=2*Pi; w[1]:=2*Pi/T;L:=T/2;w[n]:=n*w[1];
```

(1.3)

$$T := 2\pi \tag{1.3}$$

$$w_1 := 1$$

$$L := \pi$$

$$w_{n\sim} := n\sim$$

Dá-se entrada à força externa, no problema considere-se a onda quadrada

```
> f:=t->piecewise(0<t and t<T/2,1,T/2<t and t<T,-1);#periódicaf(t+
  T)=f(t)
```

$$f := t \rightarrow piecewise\left(0 < t \text{ and } t < \frac{T}{2}, 1, \frac{T}{2} < t \text{ and } t < T, -1\right) \tag{1.4}$$

```
> plot(f(t),t=0..T,thickness=2,scaling=constrained,discont=true,
  titlefont=[COURIER,DEFAULT,14],labels=["t","f(t)"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=`Gráfico
  da Entrada Periódica`);
```

Gr�fico da Entrada Peri�dica



Gráfico do prolongamento periódico da função

```
> h:=proc(t)
                                    if t>0 and t<T then f(t) elif

  t>T and t<2*T then f(t-T) elif
                                                t>-T and t<0
  then f(t+T)elif
                  t>-2*T and t<-T then f(t+2*T) fi end:
> G:=plot(h,-2*T..2*T,thickness=2,titlefont=[COURIER,DEFAULT,12],
  labels=["t", "f(t)"],labelfont=[COURIER,DEFAULT,10],axesfont=
  [COURIER,DEFAULT,10],title=` f(t)=1 se 0<t<Pi,f(t)=-1 se
  Pi<t<2Pi,T=2Pi`): G;
```



Desenvolver em série de Fourier a força externa, definida no intervalo (0,2L). Observe que se trata do **CASO II: cálculo da série de Fourier de uma função f(t) periódica, com período** T=2L, **f(t+T) = f(t), definida em um intervalo qualquer. Neste problema, c=0 e 2L=2**π.

Determinar os coeficientes a_n da série de Fourier, para n =1,2,3,...

```
> a[n]:=int(f(t)*cos(2*n*Pi*t/T),t=0..T)/L;
```

$$a_{n\sim} := 0 \tag{1.5}$$

Cálculo do coeficiente a_0

```
> a[0]:= simplify(int(f(t),t=0..T)/L);
```

$$a_0 := 0 \tag{1.6}$$

A seguir, os coeficientes b_n, para n=1,2,3,...

```
> b[n]:=int(f(t)*sin(2*n*Pi*t/T),t=0..T)/L;
```

$$b_{n\sim} := -\frac{2\left((-1)^{n\sim}-1\right)}{n\sim\,\pi} \tag{1.7}$$

A série Fourier da função é

```
> serie_f:=a[0]/2+ Sum(a[n]*cos(2*n*Pi*t/T)+b[n]*sin(2*n*Pi*t/T),n=
  1..infinity);
```

$$serie\_f := \sum_{n\sim=1}^{\infty}\left(-\frac{2\left((-1)^{n\sim}-1\right)\sin(n\sim\,t)}{n\sim\,\pi}\right) \tag{1.8}$$

Aproximar a função f(t) com a série truncada até n=3.

```
> f_ap3(t):=a[0]/2+ sum(a[n]*cos(2*n*Pi*t/T)+b[n]*sin(2*n*Pi*t/T),
  n=1..3);
```

$$f\_ap3(t) := \frac{4\sin(t)}{\pi} + \frac{4}{3}\,\frac{\sin(3\,t)}{\pi} \tag{1.9}$$

Traçar os gráficos da função dada e da sua aproximação em série de Fourier truncada.

```
> plot({f(t),f_ap3(t)},t=0..T,scaling=constrained,title=`Entrada e
  Série de Fourier truncada(n=3)`,thickness=2,color=[red,black],
  titlefont=[COURIER,DEFAULT,12],labels=["t","f(t)"],labelfont=
  [COURIER,DEFAULT,12],axesfont=[COURIER,DEFAULT,10]);
```



Aumentando o número de termos, obtém-se uma melhor aproximação. Por exemplo, para n=10 :

```
> f_ap10(t):=a[0]/2+ sum(a[n]*cos(2*n*Pi*t/T)+b[n]*sin(2*n*Pi*t/T),
  n=1..10):
> plot({f(t),f_ap10(t)},t=0..T,thickness=3, color=[red,black],
  scaling=constrained,titlefont=[COURIER,DEFAULT,12],labels=["t","f
  (t)"],labelfont=[COURIER,DEFAULT,12],axesfont=[COURIER,DEFAULT,
  10],title=`Entrada e Série de Fourier truncada(n=10)`);
```



Uma outra maneira de escrever a série Fourier da força externa é explicitando a freqëncia fundamental. A série, acima, toma a forma (observe que é o mesmo resultado).

```
> serie_forca:= a[0]/2+Sum(a[n]*cos(n*w[1]*t)+b[n]*sin(n*w[1]*t),
  'n'=1..infinity);
```

$$serie\_forca := \sum_{n=1}^{\infty} \left(- \frac{2\left((-1)^{n\sim} - 1 \right) \sin(n\sim t)}{n\sim \pi} \right) \qquad \textbf{(1.10)}$$

Parte1: b) **Forma compacta da série da força externa**. Serão calculadas a **amplitude** e a **fase,** utilizando o n-ésimo termo

```
> n_termo:=[a[n]*cos(w[n]*t)+b[n]*sin(w[n]*t)];
```

$$n\_termo := \left[-\frac{2\left(\left(-1\right)^{n\sim}-1\right)\sin\left(n\sim t\right)}{n\sim \pi} \right] \quad \textbf{(1.11)}$$

```
> amplit_forca:=root[2](a[n]^2+b[n]^2);
```

$$amplit\_forca := \frac{2\left(-\left(-1\right)^{n\sim}+1\right)\sqrt{\frac{1}{n\sim^2}}}{\pi} \quad \textbf{(1.12)}$$

Para simplificar :

```
> amplit_f_ext:=radsimp(amplit_forca);
```

$$amplit\_f\_ext := -\frac{2\left(\left(-1\right)^{n\sim}-1\right)}{n\sim \pi} \quad \textbf{(1.13)}$$

```
> if b[n]=0 then fase_forca:= Pi/2 else fase_forca_ext:=arctan(a[n]
  /b[n]) end if;
```

$$fase\_forca\_ext := 0 \quad \textbf{(1.14)}$$

```
> fext_compacta:=Sum(amplit_f_ext*sin(w[n]*t+fase_forca_ext),n=1..
  infinity);
```

$$fext\_compacta := \sum_{n\sim=1}^{\infty} \left(-\frac{2\left(\left(-1\right)^{n\sim}-1\right)\sin\left(n\sim t\right)}{n\sim \pi} \right) \quad \textbf{(1.15)}$$

ou, equivalentemente, (veja que no Maple ter-se-á o mesmo resultado, pois cos(A(+/-)Pi/2)= (-/+) sin (a))

```
> if a[n]=0 then fase_for:= Pi/2 else fase_for:=arctan(b[n]/a[n])
  end if;
> fext_compac:=Sum(amplit_f_ext*cos(w[n]*t-fase_for), n=1..
  infinity);
```

$$fase\_for := \frac{\pi}{2} \quad \textbf{(1.16)}$$

$$fext\_compac := \sum_{n\sim=1}^{\infty} \left(-\frac{2\left(\left(-1\right)^{n\sim}-1\right)\sin\left(n\sim t\right)}{n\sim \pi} \right)$$

A série com n=1..10

```
> fext_compac:=sum(amplit_f_ext*cos(w[n]*t-fase_for), n=1..10);
```

$$fext\_compac := \frac{4\sin(t)}{\pi} + \frac{4}{3}\frac{\sin(3\,t)}{\pi} + \frac{4}{5}\frac{\sin(5\,t)}{\pi} + \frac{4}{7}\frac{\sin(7\,t)}{\pi} + \frac{4}{9}\frac{\sin(9\,t)}{\pi} \quad \textbf{(1.17)}$$

```
> Graf1:=plot(fext_compac, t=0..2*Pi,color=magenta,thickness=2,
  titlefont=[COURIER,DEFAULT,12],labels=["t", "f(t)"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=`Série
  Harmônica(10 primeiros termos,1 período)`):Graf1;
```



Cálculo das amplitudes, para o espectro de freqüência correspondente à série da força externa :

```
> for i from 1 to 10 do A[i]:=evalf(subs(n=i,amplit_forca)) :  w[i]
  :=subs(n=i,w[n]):od:
> for k from 1 to 10 do lprint(frequencia[k]=w[k],amplitude[k]=A[k]
  ):od;
frequencia[1] = 1, amplitude[1] = 1.27
frequencia[2] = 2, amplitude[2] = 0.
frequencia[3] = 3, amplitude[3] = .424
frequencia[4] = 4, amplitude[4] = 0.
frequencia[5] = 5, amplitude[5] = .254
frequencia[6] = 6, amplitude[6] = 0.
frequencia[7] = 7, amplitude[7] = .182
frequencia[8] = 8, amplitude[8] = 0.
frequencia[9] = 9, amplitude[9] = .141
frequencia[10] = 10, amplitude[10] = 0.
```

Para o espectro de freqüência correspondente à série da força externa :

```
>  for i from 1 to 10 do g[i]:=line([w[i],0],[w[i],A[i]], color=
  red, thickness=3):od:
> Graf2:=plots[display]({g[1],g[2],g[3],g[4],g[5],g[6],g[7],g[8],g
  [9],g[10]},xtickmarks=[0,1,2,3,4,5,6,7,8,9,10],titlefont=
  [COURIER,DEFAULT,12],labels=["w[n]","A[n]"],labelfont=[COURIER,
  DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=`Espectro de
  Amplitude da Força,n=10`,axes=frame):Graf2;
```



Parte 2: Substituição da da Força Externa na EDO. Agora, deve-se determinar a resposta do sistema vibratório submetido à força periódica (dada acima). Para tanto, considere-se o termo geral e como dados do ploblema massa: m=1 kg, coeficiente de amortecimento do sistema: c=2 Ns/m, coef. de elasticidade da mola: k=10N/m

```
> Dados:=[m=1,c=2,k=10, f(t)=a[0]/2+Sum(n_termo,'n'=1..infinity)]:
> edmov:=subs(Dados,edo);
```

$$edmov := \frac{d^2}{dt^2}x(t) + 2\left(\frac{d}{dt}x(t)\right) + 10\,x(t) = \sum_{n\sim=1}^{\infty}\left[-\frac{2\left((-1)^{n\sim}-1\right)\sin(n\sim t)}{n\sim\,\pi}\right] \quad (1.18)$$

Parte 3: Determinar a solução da EDOLHomogênea associada x_h

```
> edmovhom:=lhs(edmov)=0;
```

$$edmovhom := \frac{d^2}{dt^2}x(t) + 2\left(\frac{d}{dt}x(t)\right) + 10\,x(t) = 0 \quad (1.19)$$

```
> base:=dsolve(edmovhom,output=basis);
```

$$base := \left[e^{-t}\sin(3\,t),\ e^{-t}\cos(3\,t)\right] \quad (1.20)$$

Tem-se um sistema forçado dissipativo e, com o último cálculo, vê-se que a solução da EDOL Homog associada é a de um sistema subamortecido

```
> solh:=dsolve(edmovhom);
```

$$solh := x(t) = \_C1\,e^{-t}\sin(3\,t) + \_C2\,e^{-t}\cos(3\,t) \quad (1.21)$$

Parte 4: Cálculo da resposta permanente x_p

Para resolver a EDOLN-H, será considerado o termo geral e o método dos coeficientes a determinar

Não é necessário usar o fator de correção, pois xh e xp não possuem termos em comum.
Para o cálculo da solução correspondente ao temo a[0]/2, deve-se supor xp=constante. Como a[0]/2=0, é óbvio que esta solução é nula. De fato, veja a seguir

```
> sup_xp0:=Cte;
```

$$sup\_xp0 := Cte \quad (1.22)$$

```
> edmov0:=lhs(edmov)=a[0]/2; #op(1,f_ap(t))
```

$$edmov0 := \frac{d^2}{dt^2}x(t) + 2\left(\frac{d}{dt}x(t)\right) + 10\,x(t) = 0 \quad (1.23)$$

```
> xp0:=solve(simplify(subs(x(t)=sup_xp0,edmov0)),Cte);
```

$$xp0 := 0 \quad (1.24)$$

Para as outras soluções serão calculados os coeficientes Ap e Bp

```
> suposta_solp:=Ap*cos(p*t)+Bp*sin(p*t);
```

$$suposta\_solp := Ap\cos(p\sim t) + Bp\sin(p\sim t) \quad (1.25)$$

Substituindo a suposta solução na E.D.

```
> calculo_coef:=radsimp(subs(x(t)=suposta_solp,edmov));
```

$$calculo\_coef := -Ap\cos(p\sim t)\,p\sim^2 - Bp\sin(p\sim t)\,p\sim^2 - 2\,Ap\sin(p\sim t)\,p\sim + 2\,Bp\cos(p\sim t)\,p\sim + 10\,Ap\cos(p\sim t) + 10\,Bp\sin(p\sim t) = \sum_{n\sim=1}^{\infty}\left[-\frac{2\left((-1)^{n\sim}-1\right)\sin(n\sim t)}{n\sim\pi}\right] \quad \textbf{(1.26)}$$

Agrupando os termos semelhantes

```
> agrup:=collect(calculo_coef,{cos(p*t),sin(p*t)});
```

$$agrup := \left(-Ap\,p\sim^2 + 2\,Bp\,p\sim + 10\,Ap\right)\cos(p\sim t) + \left(-Bp\,p\sim^2 - 2\,Ap\,p\sim + 10\,Bp\right)\sin(p\sim t) = \sum_{n\sim=1}^{\infty}\left[-\frac{2\left((-1)^{n\sim}-1\right)\sin(n\sim t)}{n\sim\pi}\right] \quad \textbf{(1.27)}$$

e identificando os coeficientes

```
> A:=(op(1,lhs(agrup)));B:=(op(2,lhs(agrup)));
```

Atenção! Conferir os coeficientes do co-seno e do seno, acima.Se a resposta não estiver correta, rodar novamente o programa.

$$A := \left(-Ap\,p\sim^2 + 2\,Bp\,p\sim + 10\,Ap\right)\cos(p\sim t) \quad \textbf{(1.28)}$$
$$B := \left(-Bp\,p\sim^2 - 2\,Ap\,p\sim + 10\,Bp\right)\sin(p\sim t)$$

```
> coef_cos:=coeffs(A,cos(p*t));coef_sen:=coeffs(B,sin(p*t));
```

$$coef\_cos := -Ap\,p\sim^2 + 2\,Bp\,p\sim + 10\,Ap \quad \textbf{(1.29)}$$
$$coef\_sen := -Bp\,p\sim^2 - 2\,Ap\,p\sim + 10\,Bp$$

Os coeficientes An e Bn são determinados resolvendo o sistema algébrico:

```
> coefp:=(solve({coef_sen=subs(n=p,b[n]), coef_cos=subs(n=p,a[n])},
  {Ap,Bp}));
```

$$coefp := \left\{Bp = \frac{2\left((-1)^{p\sim}-1\right)\left(p\sim^2-10\right)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)p\sim},\ Ap = \frac{4\left((-1)^{p\sim}-1\right)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)}\right\} \quad \textbf{(1.30)}$$

```
> xp:=(subs(coefp,suposta_solp));coeff(xp,cos(p*t));coeff(xp,sin(p*
  t));
```

$$xp := \frac{4\left((-1)^{p\sim}-1\right)\cos(p\sim t)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)} + \frac{2\left((-1)^{p\sim}-1\right)\left(p\sim^2-10\right)\sin(p\sim t)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)p\sim} \quad \textbf{(1.31)}$$

$$\frac{4\left((-1)^{p\sim}-1\right)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)}$$

$$\frac{2\left((-1)^{p\sim}-1\right)\left(p\sim^2-10\right)}{\pi\left(p\sim^4-16\,p\sim^2+100\right)p\sim}$$

```
> Amplitp:=radsimp(sqrt(coeff(xp,cos(p*t))^2+coeff(xp,sin(p*t))^2))
```

```
;
```

$$Amplitp := -\frac{2\left((-1)^{p\sim}-1\right)}{\pi\, p\sim \sqrt{p\sim^4-16\, p\sim^2+100}} \qquad (1.32)$$

```
> if coeff(xp,sin(p*t))=0 then fasep:=Pi/2 else fasep:=simplify
  (arctan(coeff(xp,cos(p*t))/(coeff(xp,sin(p*t)))))end if;
```

$$fasep := \arctan\left(\frac{2\, p\sim}{p\sim^2-10}\right) \qquad (1.33)$$

A forma compacta da p-ésima solução é dada por

```
> comp_xp:=simplify(Amplitp*sin(p*t+fasep));
```

$$comp\_xp := -\frac{2\left((-1)^{p\sim}-1\right)\sin\left(p\sim\, t+\arctan\left(\frac{2\, p\sim}{p\sim^2-10}\right)\right)}{\pi\, p\sim \sqrt{p\sim^4-16\, p\sim^2+100}} \qquad (1.34)$$

Para se obter diretamente a forma compacta das soluções: x1, x2, x3, x4, x5, x6, x7, x8, x9 e x10

```
> for i from 1 to 10 do x[i]:=evalf(subs(p=i,comp_xp)):od:for k
  from 1 to 10 do lprint(solucao[k]=x[k]):od;
solucao[1] = .138*sin(t-.218)
solucao[2] = 0.
solucao[3] = .695e-1*sin(3.*t-1.41)
solucao[4] = 0.
solucao[5] = .141e-1*sin(5.*t+.588)
solucao[6] = 0.
solucao[7] = .440e-2*sin(7.*t+.345)
solucao[8] = 0.
solucao[9] = .193e-2*sin(9.*t+.249)
solucao[10] = 0.
```

Outra maneira, para as sete primeiras soluções:

```
> x1:=evalf(subs(p=1,comp_xp));x2:=evalf(subs(p=2,comp_xp));x3:=
  evalf(subs(p=3,comp_xp));x4:=evalf(subs(p=4,comp_xp));x5:=evalf
  (subs(p=5,comp_xp));x6:=evalf(subs(p=6,comp_xp));x7:=evalf(subs
  (p=7,comp_xp));
```

$$x1 := 0.138 \sin(t-0.218) \qquad (1.35)$$

$$x2 := 0.$$

$$x3 := 0.0695 \sin(3.\, t-1.41)$$

$$x4 := 0.$$

$$x5 := 0.0141 \sin(5.\, t+0.588)$$

$$x6 := 0.$$

$$x7 := 0.00440 \sin(7.\, t+0.345)$$

Gráfico da solução x1

```
> g1:=plot(x[1],t=0..10,color=blue,titlefont=[COURIER,DEFAULT,12],
  labels=["t","x[n]"],labelfont=[COURIER,DEFAULT,10],axesfont=
  [COURIER,DEFAULT,10],title=`Gráfico da Solução x1(t)` ):display(
  {g1});
```



Para conferir o valor de x[1] em t=0. Olhe o gráfico acima!

```
> evalf(subs(t=0,x1));
```

$$-0.0298 \tag{1.36}$$

Gráfico da solução x3

```
> g3:=plot(x[3],t=0..10,color=maroon,titlefont=[COURIER,DEFAULT,
  12],labels=["t","x[n]"],labelfont=[COURIER,DEFAULT,10],axesfont=
  [COURIER,DEFAULT,10],title=`Gráfico da Solução x3(t)` ):display(
  {g3});
```



Com as dez primeiras soluções a integral particular é (lembre que xp0=0, xp2=0, xp4=0, xp6=0,xp8=0, xp10=0...)

```
> sol_xp:=sum(x[m],m=1..10);
```

$$sol\_xp := 0.138\sin(t-0.218)+0.0695\sin(3.\,t-1.41)+0.0141\sin(5.\,t+0.588)+0.00440\sin(7.\,t+0.345)+0.00193\sin(9.\,t+0.249) \tag{1.37}$$

Só para conferir o valor da resposta permanente, xp em t=0, no gráfico a seguir:

```
> evalf(subs(t=0,sol_xp));
```

$$-0.0886 \tag{1.38}$$

```
> gsol:=plot((sol_xp),t=0..10,thickness=3,titlefont=[COURIER,
  DEFAULT,12],labels=["t","x[n]"],labelfont=[COURIER,DEFAULT,10],
  axesfont=[COURIER,DEFAULT,10],title=`Gráfico da Resposta
  Permanente`):gsol;
```



E a solução geral x=xh+xp com as sete primeiras soluções é

```
> soltotal:=rhs(solh)+sol_xp;
```

$$soltotal := \_C1\, e^{-t} \sin(3\, t) + \_C2\, e^{-t} \cos(3\, t) + 0.138 \sin(t - 0.218) + 0.0695 \sin(3.\, t - 1.41) + 0.0141 \sin(5.\, t + 0.588) + 0.00440 \sin(7.\, t + 0.345) + 0.00193 \sin(9.\, t + 0.249) \tag{1.39}$$

Na seqüência, será considerada apenas a resposta permanente, pois a solução da EDOLH associada é transiente e, geralmente, não é considerada neste tipo de problema.

A seguir, as amplitudes serão calculadas (em módulo), para p=1,2,3,4,5,6,7,8,9,10 para comparar os resultados e traçar o espectro de amplitude .

```
> for i from 1 to 10 do A[i]:=evalf(abs(subs(p=i,Amplitp))) :  w[i]
  :=subs(n=i,w[n]):od:
> for m from 1 to 10 do lprint(frequencia[m]=w[m],amplitude[m]=A[m]
  ):od;
frequencia[1] = 1, amplitude[1] = .138
frequencia[2] = 2, amplitude[2] = 0.
frequencia[3] = 3, amplitude[3] = .695e-1
frequencia[4] = 4, amplitude[4] = 0.
frequencia[5] = 5, amplitude[5] = .141e-1
frequencia[6] = 6, amplitude[6] = 0.
frequencia[7] = 7, amplitude[7] = .440e-2
frequencia[8] = 8, amplitude[8] = 0.
frequencia[9] = 9, amplitude[9] = .193e-2
frequencia[10] = 10, amplitude[10] = 0.
```

Outra maneira:

```
> ap1:=abs(evalf(subs(p=1,Amplitp)));ap2:=abs(evalf(subs(p=2,
  Amplitp)));ap3:=abs(evalf(subs(p=3,Amplitp)));ap4:=abs(evalf(subs
  (p=4,Amplitp)));ap5:=abs(evalf(subs(p=5,Amplitp)));ap6:=abs(evalf
  (subs(p=6,Amplitp)));
```

$$ap1 := 0.138 \tag{1.40}$$

$$ap2 := 0.$$

$$ap3 := 0.0695$$

$$ap4 := 0.$$

$$ap5 := 0.0141$$

$$ap6 := 0.$$

Note que a amplitude dominante é a correspondente a p=1. Para p par amp=0, p ímpar, p>4, os valores serão muito pequenos.

Espectro de amplitude da resposta do sistema

```
>  for q from 1 to 10 do gr[q]:=line([w[q],0],[w[q],A[q]], color=
  red, thickness=3):od:
```

```
> Graf:=plots[display]({gr[1],gr[2],gr[3],gr[4],gr[5],gr[6],gr[7],
  gr[8],gr[9],gr[10]},xtickmarks=[0,1,2,3,4,5,6,7,8,9,10],
  titlefont=[COURIER,DEFAULT,12],labels=["w[n]","A[n]"],labelfont=
  [COURIER,DEFAULT,10],axesfont=[COURIER,DEFAULT,10],title=
  `Espectro de Amplitude da Saída,n=10`,axes=frame):Graf;
```



A forma compacta da resposta (saída) do sistema é dada por

$$saida = \sum_{n=1}^{\infty} [Amp\_entrada]\,[Fat\_ganho]\,\sin\left(n\,\omega_n\,t + \Phi_n + \psi_n\right)$$, onde Φ_n denota a fase da resposta e ψ_n a fase da força externa.

O Fator de Ganho ou Fator deslocamento (vamos usar letras maiúsculas para massa M, coeficiente de amortecimento C e coeficiente de elasticidade K, para não gerar confusão, após tantos índices usados) é calculado como

```
> Fator_des:=1/(sqrt((C^2*N^2*w[n]^2)+(K-M*N^2*w[n]^2)^2));F:=subs(
```

```
(M=1,C=2,K=10),Fator_des);
```

$$Fator\_des := \frac{1}{\sqrt{C^2 N^2 n\sim^2 + K^2 - 2 K M N^2 n\sim^2 + M^2 N^4 n\sim^4}} \quad \textbf{(1.41)}$$

$$F := \frac{1}{\sqrt{-16 N^2 n\sim^2 + 100 + N^4 n\sim^4}}$$

```
> saida:=a[0]/2+Sum(amplit_f_ext*Fator_des)*sin(w[n]*t+Phi[n]+psi[n]);
```

$$saida := \left(\sum -\frac{2\left((-1)^{n\sim} - 1\right)}{n\sim \pi \sqrt{C^2 N^2 n\sim^2 + K^2 - 2 K M N^2 n\sim^2 + M^2 N^4 n\sim^4}} \right) \sin\left(n\sim t + \Phi_{n\sim} + \psi_{n\sim}\right) \quad \textbf{(1.42)}$$

Agora, **para conferir o resultado:** neste problema, a maior amplitude da força externa é a primeira, então calcula-se a aplitude do primeiro termo da força multiplicada pelo fator de ganho

```
> amp_max:=evalf(subs((n=1,N=1),F)*subs(n=1,sqrt(a[n]^2+b[n]^2)));
```

$$amp\_max := 0.138 \quad \textbf{(1.43)}$$

```
> Amplit1 := evalf(subs(p=1,Amplitp));
```

$$Amplit1 := 0.138 \quad \textbf{(1.44)}$$

Conclusão: A resposta permanente do sistema (xp(t)) é dominada pela solução xp1(t), que corresponde ao primeiro termo da série de Fourier da resposta e cuja freqüência (w=1) é a frequência fundamental.